

Let's Give 'Em Something to TOC about: Transforming the Table of Contents of Your PDF File

Bari Lawhorn, SAS Technical Support
SAS Institute Inc., Cary, NC

Where Can We TOC?


The following SAS Output Delivery System (ODS) destinations create tables of contents (TOCs):

- HTML
- RTF
- EXCEL
- TAGSETS.EXCELXP
- WORD (pre-production)
- PDF

TOC This Way: The ODS HTML Destination

The HTML destination creates a TOC when you include the CONTENTS= option in the ODS HTML statement:


```
ods html file="file.html"  
        contents="contents.html" (title="HTML  
                                TOC");
```



TOC This Way: The ODS RTF Destination

The RTF destination creates a table of contents when you add the TOC_DATA and CONTENTS arguments to the ODS RTF statement:

```
ods rtf file="file.rtf" contents toc_data;
```



TOC This Way: The ODS EXCEL Destination

The Excel destination creates a TOC when you use `OPTIONS(CONTENTS="YES")`:


```
ods excel file="file.xlsx"  
options (contents="yes");
```

	A	B
1	The Print Procedure	
2	Data Set SASHELP.CARS	
3	The SGPlot Procedure	
4	The SGPlot Procedure	
5		

TOC This Way: The ODS PDF Destination

The PDF destination creates a table of contents by default.

```
ods pdf file="file.pdf";
```



The screenshot displays the SAS user interface. On the left, a 'Bookmarks' panel is open, showing a list of procedures: 'The Print Procedure' (expanded to show 'Data Set SASHELP.CARS'), 'The SGPlot Procedure' (expanded to show 'The SGPlot Procedure'), and another 'The SGPlot Procedure'. On the right, a data table is displayed with the following columns: Obs, Make, Model, Type, Origin, DriveTrain, MSRP, Invoice, and EngineSize. The table contains 8 rows of data.

Obs	Make	Model	Type	Origin	DriveTrain	MSRP	Invoice	EngineSize
1	Acura	MDX	SUV	Asia	All	\$36,945	\$33,337	3.5
2	Acura	RSX Type S 2dr	Sedan	Asia	Front	\$23,820	\$21,761	2.0
3	Acura	TSX 4dr	Sedan	Asia	Front	\$26,990	\$24,647	2.4
4	Acura	TL 4dr	Sedan	Asia	Front	\$33,195	\$30,299	3.2
5	Acura	3.5 RL 4dr	Sedan	Asia	Front	\$43,755	\$39,014	3.5
6	Acura	3.5 RL w/Navigation 4dr	Sedan	Asia	Front	\$46,100	\$41,100	3.5
7	Acura	NSX coupe 2dr manual S	Sports	Asia	Rear	\$89,765	\$79,978	3.2
8	Audi	A4 1.8T 4dr	Sedan	Europe	Front	\$25,940	\$23,508	1.8

More TOC with the ODS PDF Destination

To create a **printable** table of contents, use the CONTENTS=YES option:

```
ods pdf file="file.pdf" contents=yes;
```

Table of Contents	
The Print Procedure	1
Data Set SASHELP.CARS	1
The SGPlot Procedure	25
The SGPlot Procedure	25

Is the TOC Flexible?

Yes! You can change the following elements in the bookmarks:

- the node text
- the node location
- the node generation

Shop TOC: Making Your TOC Meaningful

To TOC more meaningfully, you can use the following SAS elements:

- ODS PROCLABEL statement
- CONTENTS=, DESCRIPTION=, and OBJECTLABEL= arguments
- DOCUMENT destination and procedure
- NOBOOKMARKGEN argument

Shop TOC: The First Node


Every SAS procedure and DATA step generates at least two nodes, as shown in the following example:

```
ods pdf file="file.pdf";

proc print data=sashelp.cars;
run;

proc sgplot data=sashelp.cars;
  vbar type;
run;


ods pdf close;
```



Shop TOC: Changing the Text of the First Node

The ODS PROCLABEL statement controls the first node.

```
ods pdf file="proclabel.pdf";  
ods proclabel="Cars from the USA";  
proc print data=sashelp.cars;  
  where origin="USA";run;  
ods proclabel=  
"Vertical Bar chart of Car Type";  
proc sgplot data=sashelp.cars;  
  where origin="USA";  
  vbar type;run;  
ods pdf close;
```



Shop TOC: Controlling the Second Node

Option 1: You can hide it!

```
ods pdf file="proclabel.pdf" pdftoc=1;
```

```
ods proclabel="Cars from the USA";
```

```
proc print data=sashelp.cars;
```

```
  where origin="USA";
```

```
run;
```

```
ods proclabel=
```

```
"Vertical Bar chart of Car Type";
```


```
proc sgplot data=sashelp.cars;
```

```
  where origin="USA";
```

```
  vbar type;
```

```
run;
```

```
ods pdf close;
```



Shop TOC: Controlling the Second Node

(continued)

Option 2: You can change it!

- Use the CONTENTS= argument in the PRINT, REPORT, and TABULATE procedures.
- Use the DESCRIPTION= argument in ODS GRAPHICS and SAS/GRAPH® software procedures.
- Use the OBJECTLABEL= argument in the DATA step and in the FILE PRINT ODS= statement.

Shop TOC: Controlling the Second Node with the CONTENTS= argument

Option 2: Changing the second node with the CONTENTS= argument in PROC PRINT

```
ods pdf file="first_second_node.pdf";
```


```
ods proclabel="Cars from the USA";
```

```
proc print data=sashelp.cars  
  contents="First table";
```

```
  where origin="USA";
```

```
run;
```

```
ods pdf close;
```



Shop TOC: Controlling the Second Node with the CONTENTS= argument (continued)


You can use the CONTENTS= argument in these three procedures:

- PROC PRINT
- PROC TABULATE*
- PROC REPORT*

*You can also control the third node of these procedures with CONTENTS= on statements inside these procedures


Shop TOC: Controlling the Second and Third Node with the CONTENTS= argument

```
ods proclabel="Cars from the USA";  
proc report data=sashelp.cars contents="First table";  
where origin="USA";  
define origin / group;  
break before origin / page contents="Make, Model, Type, etc.";  
run;  
ods proclabel="Highway MPG Mean";  
proc tabulate data=sashelp.cars  
    contents="Second table";  
    class origin ;var mpg_highway;  
    table origin, mpg_highway*(mean) /  
        contents="by Origin";  
run;
```



Shop TOC: Controlling the Second Node with the DESCRIPTION= argument

```
ods proclabel="Histogram of MPG_City";  
proc sgpanel data=sashelp.cars  
  description="Paneled by Type";  
  panelby type / rows=2 columns=3;  
  histogram mpg_city;  
run;  
ods proclabel="Plot of SASHELP.CARS";  
proc gplot data=sashelp.cars;  
  plot type*horsepower /  
    description="Type by Horsepower";  
run; quit;
```




Shop TOC: Controlling the Second Node with the OBJECTLABEL= argument

```
ods pdf file="objectlabel.pdf";


ods proclabel='DATA step we love you';
data _null_;
  file print ods=(objectlabel="Complete Cars Table");
  set sashelp.cars;
  put _ods_;
run;

ods pdf close;
```



Big TOCers: Using a Procedure That Creates Many Nodes

```
ods pdf file="univariate.pdf";  
  
ods proclabel='Hybrid vehicles';  
proc univariate data=sashelp.cars;  
  where type="Hybrid";  
  var weight;  
run;  
  
ods pdf close;
```



Shop TOC: Using the ODS TRACE ON Statement to Determine Which Tables Are Used

```
ods trace on;
```

```
ods pdf file="univariate.pdf" ;  
  proc univariate data=sashelp.cars;  
    where type="Hybrid";  
    var weight;  
run;
```

```
ods pdf close;
```

LOG

Window

```
Output Added:  
-----
```

```
Name:      Moments  
Label:     Moments  
Template:  base.univariate.Moments  
Path:     Univariate.Weight.Moments  
-----
```


Shop TOC: Creating a Macro Variable for the CONTENTS_LABEL= Attribute

```
proc template;  
  edit base.univariate.moments;  
    mvar momentscl;  
    contents_label=Momentscl;  
  end;  
run;
```

```
%let Momentscl=Hybrid Moments Table;  
ods pdf file="univariate_edited.pdf";
```

```
ods proclabel='Hybrid vehicles';  
proc univariate data=sashelp.cars;  
  where type="Hybrid";  
  var weight;  
run;  
ods pdf close;
```

Use PROC TEMPLATE to create a macro variable for the CONTENTS_LABEL= attribute.



Shop TOC: Other Methods for Controlling the Second Node

Option 3. You can eliminate it!

Use a null value for the following arguments to eliminate the node. A *null value* means there is no space between the quotation marks:

- CONTENTS=""
- OBJECTLABEL=""

Shop TOC: Other Methods for Controlling the Second Node (continued)

```
ods pdf file="null.pdf";

ods proclabel='Single DATA step TOC node';
data _null_;
  file print ods=(objectlabel="");
  set sashelp.cars;
  put _ods_;
run;
```

- Single DATA step TOC node
- PRINT with 1 TOC node


```
ods proclabel="PRINT with 1 TOC node";
proc print data=sashelp.cars contents='';
run;
```

```
ods pdf close;
```

Shop TOC: Showing Only One Set of Nodes per Step


To eliminate all nodes for a particular procedure or DATA step, use the NOBOOKMARKGEN argument.

```
ods pdf file="onepage.pdf" columns=2;
ods proclabel="Histogram of Horsepower";
proc sgpanel data=sashelp.cars description="Panel by
type";
    panelby type / rows=2 columns=3;
    histogram horsepower;run;
ods pdf nobookmarkgen;
proc print data=sashelp.cars;
run;
ods pdf close;
```




Using ODS and the DOCUMENT Procedure with More Complicated TOCs

You have this TOC:




You WANT this TOC:



Using ODS and the DOCUMENT Procedure with More Complicated TOCs (continued)

```
ods document name=nested(write);  
  
ods pdf file="before.pdf";  
  proc sql;  
    select * from sashelp.cars;  
  run;  
quit;  
  
ods select parameterestimates;  
proc reg data=sashelp.cars ;  
  model mpg_highway=mpg_city;  
run;  
quit;  
  
ods pdf close;  
ods document close;
```



Using ODS and the DOCUMENT Procedure with More Complicated TOCs (continued)

```
proc document name=nested;  
  list / levels=all;  
run;  
quit;
```

Listing of: \Work.Nested\		
Order by: Insertion		
Number of levels: All		
Obs	Path	Type
1	\SQL#1	Dir
2	\SQL#1\SQL_Results#1	Table
3	\Reg#1	Dir
4	\Reg#1\MODEL1#1	Dir
5	\Reg#1\MODEL1#1\Fit#1	Dir
6	\Reg#1\MODEL1#1\Fit#1\MPG_Highway#1	Dir
7	\Reg#1\MODEL1#1\Fit#1\MPG_Highway#1\ParameterEstimates#1	Table

Using ODS and the DOCUMENT Procedure with More Complicated TOCs (continued)

```
proc document name=nested;
```

Listing of: \Work.Nested\

Order by: Insertion

Number of levels: All


Obs	Path	Type	Size in Bytes	Created	Modified	Symbolic Link	Template	Label
1	\SQL#1	Dir		18MAY2023:14:15:24	18MAY2023:14:15:24			
2	\SQL#1\SQL_Results#1	Table	22235	18MAY2023:14:15:24	18MAY2023:14:15:24		Base.SQL	Child 1
3	\Reg#1	Dir		18MAY2023:14:15:24	18MAY2023:14:15:24			The Reg Procedure
4	\Reg#1\MODEL1#1	Dir		18MAY2023:14:15:24	18MAY2023:14:15:24			MODEL1
5	\Reg#1\MODEL1#1\Fit#1	Dir		18MAY2023:14:15:24	18MAY2023:14:15:24			Fit
6	\Reg#1\MODEL1#1\Fit#1\MPG_Highway#1	Dir		18MAY2023:14:15:24	18MAY2023:14:15:24			
7	\Reg#1\MODEL1#1\Fit#1\MPG_Highway#1\ParameterEstimates#1	Table	325	18MAY2023:14:15:24	18MAY2023:14:15:24		Stat.REG.ParameterEstimates	Child 2
8	\first#1	Dir		18MAY2023:14:15:24	18MAY2023:14:15:24			Parent
9	\first#1\SQL_Results#1	Table	22235	18MAY2023:14:15:24	18MAY2023:14:15:24		Base.SQL	Child 1
10	\first#1\ParameterEstimates#1	Table	325	18MAY2023:14:15:24	18MAY2023:14:15:24		Stat.REG.ParameterEstimates	Child 2

```
QUIT;
```

```
run;
```

Using ODS and the DOCUMENT Procedure with More Complicated TOCs (continued)

```
proc document name=nested;  
ods pdf file="after.pdf";  
  replay \first#1;  
run;  
ods pdf close;  
quit;
```



Don't TOC Ugly!

Certain complications can occur that might frustrate you. Before you call SAS Technical Support, check for the following situations:


- Complication 1: None of my changes are visible.
- Complication 2: My BY variables are showing.
- Complication 3: My inline-style commands are showing.
- Complication 4: The second node is blank.
- Complication 5: All my bookmarks are showing in my tagged PDF file!

Complication 1: None of My Changes Are Showing

Problem:

```
options label;  
ods pdf file="nolabel.pdf";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods proc label="Plots of SASHELP.CARS";  
ods pdf close;  
ods pdf close;
```


```
4 %put %sysfunc(getoption(label));  
NOLABEL
```



Complication 2: My BY Variables Are Showing

Problem:

```
ods pdf file="byvars.pdf";  
  
proc sort data=sashelp.cars out=sorted;  
  by origin;  
ods proclabel="Plots of SASHELP.CARS";  
proc sgplot data=sorted  
  description="Vertical Bar of Type";  
  by origin;  
  vbar type;  
run;  
  
ods pdf close;
```




Solution: Use ODS and PROC DOCUMENT or MACRO logic.

Complication 3: My Inline-Style Commands Are Showing

Problem:

```
ods escapechar="^";  
ods pdf file="inline_styles.pdf";  
  
proc print data=sashelp.cars  
contents="^S={foreground=red}Red^S={foreground=blue}Blue";  
run;  
  
ods pdf close;
```



Self TOC: Love and accept the font and color of your TOC nodes.

Complication 4: The Second Node Is Blank

PROBLEM:

```
ods pdf file="blank_vs_null.pdf";
```

```
proc print data=sashelp.cars contents=" ";  
run;
```

```
ods pdf close;
```




Solution: Pay attention to the spacing between your quotation marks.

Complication 5: All my bookmarks are showing in my tagged PDF file

```
ods pdf file="proclabel.pdf" accessible pdftoc=1;
ods proclabel="Cars from the USA";
proc print data=sashelp.cars (where=(origin="USA"));
run;
ods proclabel=
"Vertical Bar chart of Car Type";
proc sgplot data=sashelp.cars (
  where=(origin="USA"));
  vbar type;run;
ods pdf close;
```

TOC Inclusively: Don't try to hide bookmarks from screen readers



Smooth TOCin'

Now that you know how to use the following SAS elements, you can TOC more meaningfully!

- ODS PROCLABEL statement
- CONTENTS=, DESCRIPTION=, and OBJECTLABEL= arguments
- TEMPLATE procedure
- DOCUMENT destination and procedure
- NOBOOKMARKGEN argument

With these elements in your repertoire, you'll really leave your audience with something to TOC about!

Resources

- [51626 - How to write the contents of the SAS® log or other text output to an HTML, RTF, or PDF file](#)
- [24064 - In ODS HTML, PDF, and RTF, how can I change the text specified at the top of the table of contents or table of pages?](#)
- [Customize Your Table of Contents with the ODS Destination for Word](#)
- [Have It Your Way: Rearrange and Replay Your Output with ODS DOCUMENT](#)
- [PROC DOCUMENT by Example Using SAS - Michael Tuchman - Google Books](#)
- [The Incredible, Accessible Report: Creating PDF Reports that Meet Compliance Standards in SAS® 9.4M6](#)
- [Let's Give 'Em Something to TOC about: Transforming the Table of Contents of Your PDF File](#)



TOC Well!

sas.com

